



HOME CREDIT



Virtual Internship Experience

Data Processing using Excel

IF and IFS Function

SUM, SUMIF, and SUMPRODUCT

Statistics-related Function (MIN, MAX, MEDIAN, etc)

VLOOKUP & HLOOKUP

1. IF AND IFS FUNCTION

● If Function

The IF function is one of the most popular functions in Excel, and it allows you to make logical comparisons between a value and what you expect. So an IF statement can have two results. The first result is if your comparison is True, the second if your comparison is False.

For example,

=IF(C2="Yes",1,2) says IF(C2 = Yes, then return a 1, otherwise return a 2).

Syntax

Use the IF function, one of the logical functions, to return one value if a condition is true and another value if it's false.

IF(logical_test, value_if_true, [value_if_false])

For example:

=IF(A2>B2,"Over Budget","OK")

=IF(A2=B2,B4-A4,"")

Argument name	Description
logical_test (required)	The condition you want to test.
value_if_true (required)	The value that you want returned if the result of logical_test is TRUE.
value_if_false (optional)	The value that you want returned if the result of logical_test is FALSE.

Simple IF examples

<i>f_x</i>	=IF(C2="Yes",1,2)	
	C	D
	Active?	Activity Code
	Yes	1

Simple IF examples

=IF(C2="Yes",1,2)

In the above example, cell D2 says: IF(C2 = Yes, then return a 1, otherwise return a 2)

<i>f_x</i>	=IF(C2=1,"Yes","No")	
	C	D
	Active?	Activity Code
	1	Yes

=IF(C2=1,"Yes","No")

In this example, the formula in cell D2 says: IF(C2 = 1, then return Yes, otherwise return No)As you see, the IF function can be used to evaluate both text and values. It can also be used to evaluate errors. You are not limited to only checking if one thing is equal to another and returning a single result, you can also use mathematical operators and perform additional calculations depending on your criteria. You can also nest multiple IF functions together in order to perform multiple comparisons.

=IF(C2>B2,"Over Budget", "Within Budget")			
B	C	D	E
Budgeted	Actual	Status	Amount Over
\$800.00	\$921.58	Over Budget	\$121.58
\$375.00	\$324.98	Within Budget	\$0.00
\$150.00	\$128.43	Within Budget	\$0.00
\$150.00	\$174.38	Over Budget	\$24.38

=IF(C2>B2,"Over Budget","Within Budget")

In the above example, the IF function in D2 is saying IF(C2 Is Greater Than B2, then return "Over Budget", otherwise return "Within Budget")

=IF(C2>B2,C2-B2,0)			
B	C	D	E
Budgeted	Actual	Status	Amount Over
\$800.00	\$921.58	Over Budget	\$121.58
\$375.00	\$324.98	Within Budget	\$0.00
\$150.00	\$128.43	Within Budget	\$0.00
\$150.00	\$174.38	Over Budget	\$24.38

=IF(C2>B2,C2-B2,0)

In the above illustration, instead of returning a text result, we are going to return a mathematical calculation. So the formula in E2 is saying IF(Actual is Greater than Budgeted, then Subtract the Budgeted amount from the Actual amount, otherwise return nothing).

=IF(E7="Yes",F5*0.0825,0)			
C	D	E	F
Item	Quantity	Cost	Total
Widget	2	\$2.90	\$5.80
Doochickey	3	\$8.55	\$25.66
	Sub-Total	\$11.45	\$31.46
	Sales Tax?	Yes	\$2.60
	Total		\$34.05

=IF(E7="Yes",F5*0.0825,0)

In this example, the formula in F7 is saying IF(E7 = "Yes", then calculate the Total Amount in F5 * 8.25%, otherwise no Sales Tax is due so return 0)

Note: If you are going to use text in formulas, you need to wrap the text in quotes (e.g. "Text"). The only exception to that is using TRUE or FALSE, which Excel automatically understands.

Common problems

Problem	What went wrong
0 (zero) in cell	There was no argument for either value_if_true or value_if_False arguments. To see the right value returned, add argument text to the two arguments, or add TRUE or FALSE to the argument.
#NAME? in cell	This usually means that the formula is misspelled.

• IFS function

The IFS function checks whether one or more conditions are met, and returns a value that corresponds to the first TRUE condition. IFS can take the place of

multiple nested IF statements, and is much easier to read with multiple conditions.

Simple syntax

Generally, the syntax for the IFS function is:

=IFS([Something is True1, Value if True1, Something is True2, Value if True2, Something is True3, Value if True3])

Please note that the IFS function allows you to test up to 127 different conditions. However, we don't recommend nesting too many conditions with IF or IFS statements. This is because multiple conditions need to be entered in the correct order, and can be very difficult to build, test and update.

Syntax

IFS(logical_test1, value_if_true1, [logical_test2, value_if_true2], [logical_test3, value_if_true3],...)

Argument	Description
logical_test1 (required)	Condition that evaluates to TRUE or FALSE.
value_if_true1 (required)	Result to be returned if logical_test1 evaluates to TRUE. Can be empty.
logical_test2... logical_test127 (optional)	Condition that evaluates to TRUE or FALSE.
value_if_true2... value_if_true127 (optional)	Result to be returned if logical_testN evaluates to TRUE. Each value_if_trueN corresponds with a condition logical_testN . Can be empty.

Example 1

B2 =IFS(A2>89,"A",A2>79,"B",A2>69,"C",A2>59,"D",TRUE,"F")			
	A	B	C
1	Grade	Letter	Result
2	93	A	"A", because A2>89
3	89	B	"B", because B3>79
4	71	C	"C", because B4>69
5	60	D	"D", because A5>59
6	58	F	"F", because 58 doesn't meet the prior conditions. "TRUE" and its corresponding value "F" provide a default value because the other conditions aren't met.
7			

The formula for cells A2:A6 is:

=IFS(A2>89,"A",A2>79,"B",A2>69,"C",A2>59,"D",TRUE,"F")

Which says IF(A2 is Greater Than 89, then return a "A", IF A2 is Greater Than 79, then return a "B", and so on and for all other values less than 59, return an "F").

Example 2

fx =IFS(F2=1,D2,F2=2,D3,F2=3,D4,F2=4,D5,F2=5,D6,F2=6,D7,F2=7,D8)						
C	D	E	F	G	H	
	Day of the Week		Day Number	Day		
	Sunday		3	Tuesday		
	Monday					
	Tuesday					
	Wednesday					
	Thursday					
	Friday					
	Saturday					

Which says IF(A2 is Greater Than 89, then return a "A", IF A2 is Greater Than 79, then return a "B", and so on and for all other values less than 59, return an "F").

The formula in cell G7 is:

=IFS(F2=1,D2,F2=2,D3,F2=3,D4,F2=4,D5,F2=5,D6,F2=6,D7,F2=7,D8)

Which says IF(the value in cell F2 equals 1, then return the value in cell D2, IF the value in cell F2 equals 2, then return the value in cell D3, and so on, finally ending with the value in cell D8 if none of the other conditions are met).

Remarks

To specify a default result, enter TRUE for your final logical_test argument. If none of the other conditions are met, the corresponding value will be returned. In Example 1, rows 6, and 7 (with the 58 grade) demonstrate this.

- If a logical_test argument is supplied without a corresponding value_if_true, this function shows a "You've entered too few arguments for this function" error message.
- If a logical_test argument is evaluated and resolves to a value other than TRUE or FALSE, this function returns a #VALUE! error.
- If no TRUE conditions are found, this function returns #N/A error.

2. SUM, SUMIF, and SUMPRODUCT

● SUM function

The SUM function adds values. You can add individual values, cell references or ranges or a mix of all three.

For example:

=SUM(A2:A10) Adds the values in cells A2:10.

=SUM(A2:A10, C2:C10) Adds the values in cells A2:10, as well as cells C2:C10.

Syntax:

SUM(number1,[number2],...)

Argument name	Description
number1 Required	The first number you want to add. The number can be like 4, a cell reference like B6, or a cell range like B2:B8.
number2-255 Optional	This is the second number you want to add. You can specify up to 255 numbers in this way.

Best Practices with SUM

This section will discuss some best practices for working with the SUM function. Much of this can be applied to working with other functions as well.

The =1+2 or =A+B Method – While you can enter =1+2+3 or =A1+B1+C2 and get fully accurate results, these methods are error prone for several reasons:

1. Typos – Imagine trying to enter more and/or much larger values like this:

=14598.93+65437.90+78496.23

Then try to validate that your entries are correct. It's much easier to put these values in individual cells and use a SUM formula. In addition, you can format the values when they're in cells, making them much more readable than when they're in a formula.



<i>fx</i>	=SUM(D2:D4)
	D
	Data
	\$14,598.93
	\$65,437.90
	\$78,496.23
	\$158,533.06

Then try to validate that your entries are correct. It's much easier to put these values in individual cells and use a SUM formula. In addition, you can format the values when they're in cells, making them much more readable than when they're in a formula.

2. #VALUE! errors from referencing text instead of numbers

If you use a formula like:

=A1+B1+C1 or =A1+A2+A3

D2		:	  <i>fx</i>	=A2+B2+C2	
	A	B	C	D	E
1	Data 1	Data 2	Data 3	=A+B+C	
2	1	A	3	#VALUE!	
3					

Your formula can break if there are any non-numeric (text) values in the referenced cells, which will return a #VALUE! error. SUM will ignore text values and give you the sum of just the numeric values.

D2	:	x	✓	<i>f_x</i>	=SUM(A2:C2)
	A	B	C	D	
1	Data 1	Data 2	Data 3	SUM	
2	1	A	3	4	
3					

3. #REF! error from deleting rows or columns

C2	:	x	✓	<i>f_x</i>	=A2+#REF!+B2
	A	B	C	D	E
1	Data 1	Data 3	=A+B+C		
2	1	3	#REF!		
3					

If you delete a row or column, the formula will not update to exclude the deleted row and it will return a #REF! error, where a SUM function will automatically update.

C2	:	x	✓	<i>f_x</i>	=SUM(A2:B2)
	A	B	C	D	
1	Data 1	Data 3	SUM		
2	1	3	4		
3					

4. Formulas won't update references when inserting rows or columns

E2	:	x	✓	<i>f_x</i>	=A2+B2+D2
	A	B	C	D	E
1	Data 1	Data 2	Inserted	Data 3	=A+B+C
2	1	2		3	6
3					

If you insert a row or column, the formula will not update to include the added row, where a SUM function will automatically update (as long as you're not outside of the range referenced in the formula). This is especially important if you expect your formula to update and it doesn't, as it will leave you with

incomplete results that you might not catch.

E2					=SUM(A2:D2)
	A	B	C	D	E
1	Data 1	Data 2	Inserted	Data 3	SUM
2	1	3		3	7
3					

SUM with individual Cell References vs. Ranges

Using a formula like:

=SUM(A1,A2,A3,B1,B2,B3)

Is equally error prone when inserting or deleting rows within the referenced range for the same reasons. It's much better to use individual ranges, like:

=SUM(A1:A3,B1:B3)

Which will update when adding or deleting rows.

● SUMIF Function

You use the SUMIF function to sum the values in a range that meet criteria that you specify. For example, suppose that in a column that contains numbers, you want to sum only the values that are larger than 5. You can use the following formula: =SUMIF(B2:B25,">5")

Tips:

- If you want, you can apply the criteria to one range and sum the corresponding values in a different range. For example, the formula =SUMIF(B2:B5, "John", C2:C5) sums only the values in the range C2:C5, where the corresponding cells in the range B2:B5 equal "John."
- To sum cells based on multiple criteria, see [SUMIFS function](#).

Important: The SUMIF function returns incorrect results when you use it to match strings longer than 255 characters or to the string #VALUE!.

Syntax

SUMIF(range, criteria, [sum_range])

The SUMIF function syntax has the following arguments:

range Required. The range of cells that you want evaluated by criteria. Cells in each range must be numbers or names, arrays, or references that contain numbers. Blank and text values are ignored. The selected range may contain dates in standard Excel format (examples below).

criteria Required. The criteria in the form of a number, expression, a cell reference, text, or a function that defines which cells will be added. Wildcard characters can be included – a question mark (?) to match any single character, an asterisk (*) to match any sequence of characters. If you want to find an actual question mark or asterisk, type a tilde (~) preceding the character.

For example, criteria can be expressed as 32, ">32", B5, "3?", "apple*", "*~?", or TODAY().

Important: Any text criteria or any criteria that includes logical or mathematical symbols must be enclosed in double quotation marks ("). If the criteria is numeric, double quotation marks are not required.

sum_range Optional. The actual cells to add, if you want to add cells other than those specified in the range argument. If the sum_range argument is omitted, Excel adds the cells that are specified in the range argument (the same cells to which the criteria is applied).

Sum_range should be the same size and shape as range. If it isn't, performance may suffer, and the formula will sum a range of cells that starts with the first cell in sum_range but has the same dimensions as range. For example:

<i>range</i>	<i>sum_range</i>	Actual summed cells
A1:A5	B1:B5	B1:B5
A1:A5	B1:K5	B1:B5

Example 1

Copy the example data in the following table, and paste it in cell A1 of a new Excel

worksheet. For formulas to show results, select them, press F2, and then press Enter. If you need to, you can adjust the column widths to see all the data.

Property Value	Commission	Data
\$100,000	\$7,000	\$250,000
\$200,000	\$14,000	
\$300,000	\$21,000	
\$400,000	\$28,000	

Formula	Description	Result
=SUMIF(A2:A5,">160000",B2:B5)	Sum of the commissions for property values over \$160,000.	\$63,000
=SUMIF(A2:A5,">160000")	Sum of the property values over \$160,000.	\$900,000
=SUMIF(A2:A5,300000,B2:B5)	Sum of the commissions for property values equal to \$300,000.	\$21,000
=SUMIF(A2:A5,">" & C2,B2:B5)	Sum of the commissions for property values greater than the value in C2.	\$49,000

Example 2

Copy the example data in the following table, and paste it in cell A1 of a new Excel worksheet. For formulas to show results, select them, press F2, and then press Enter. If you need to, you can adjust the column widths to see all the data.

Category	Food	Sales
Vegetables	Tomatoes	\$2,300
Vegetables	Celery	\$5,500
Fruits	Oranges	\$800
	Butter	\$400
Vegetables	Carrots	\$4,200
Fruits	Apples	\$1,200

Formula	Description	Result
=SUMIF(A2:A7,"Fruits",C2:C7)	Sum of the sales of all foods in the "Fruits" category.	\$2,000
=SUMIF(A2:A7,"Vegetables",C2:C7)	Sum of the sales of all foods in the "Vegetables" category.	\$12,000
=SUMIF(B2:B7,"*es",C2:C7)	Sum of the sales of all foods that end in "es" (Tomatoes, Oranges, and Apples).	\$4,300
=SUMIF(A2:A7,"",C2:C7)	Sum of the sales of all foods that do not have a category specified.	\$400

● SUMPRODUCT function

The SUMPRODUCT function returns the sum of the products of corresponding ranges or arrays. The default operation is multiplication, but addition, subtraction, and division are also possible.

In this example, we'll use SUMPRODUCT to return the total sales for a given item and size:

D10 =SUMPRODUCT((B2:B7=B10)*(C2:C7=C10)*D2:D7)

	A	B	C	D	E	F
1		Item	Size	Sold		
2		X	S	45		
3		Y	M	21		
4		Z	L	25		
5		X	L	20		
6		Y	M	41		
7		Z	S	19		
8						
9		Item	Size	Total		
10		Y	M	62		
11						

SUMPRODUCT matches all instances of Item Y/Size M and sums them, so for this example 21 plus 41 equals 62.

Syntax

To use the default operation (multiplication):

=SUMPRODUCT(array1, [array2], [array3], ...)

The SUMPRODUCT function syntax has the following arguments:

Argument	Description
array1 Required	The first array argument whose components you want to multiply and then add.
[array2], [array3],... Optional	Array arguments 2 to 255 whose components you want to multiply and then add.

To perform other arithmetic operations

Use SUMPRODUCT as usual, but replace the commas separating the array arguments with the arithmetic operators you want (*, /, +, -). After all the operations are performed, the results are summed as usual.

Note: If you use arithmetic operators, consider enclosing your array arguments in parentheses, and using parentheses to group the array arguments to control the order of arithmetic operations.

Remarks

The array arguments must have the same dimensions. If they do not, SUMPRODUCT returns the #VALUE! error value. For example, =SUMPRODUCT(C2:C10,D2:D5) will return an error since the ranges aren't the same size.

SUMPRODUCT treats non-numeric array entries as if they were zeros.

For best performance, SUMPRODUCT should not be used with full column references. Consider =SUMPRODUCT(A:A,B:B), here the function will multiply the 1,048,576 cells in column A by the 1,048,576 cells in column B before adding them.

Example 1

D7	X ✓ f_x	=SUMPRODUCT(C2:C5,D2:D5)			
	A	B	C	D	
1		Item	Cost per Unit	Quantity	
2		Green Tea	\$3.25	9	
3		Chai	\$2.20	7	
4		Mint	\$4.20	3	
5		Ginger	\$3.62	6	
6					
7			Total Sales	\$78.97	
8					

To create the formula using our sample list above, type =SUMPRODUCT(C2:C5,D2:D5) and press Enter. Each cell in column C is multiplied by its corresponding cell in the same row in column D, and the results are added up. The total amount for the groceries is \$78.97.

To write a longer formula that gives you the same result, type

=C2*D2+C3*D3+C4*D4+C5*D5 and press Enter. After pressing Enter, the result is the same: \$78.97. Cell C2 is multiplied by D2, and its result is added to the result of cell C3 times cell D3 and so on.

Example 2

The following example uses SUMPRODUCT to return the total net sales by sales agent, where we have both total sales and expenses by agent. In this case, we're using an Excel table, which uses structured references instead of standard Excel ranges. Here you'll see that the Sales, Expenses, and Agent ranges are referenced by name.

C8		=SUMPRODUCT(((Table1[Sales])+Table1[Expenses]))*(Table1[Agent]=B8))						
	A	B	C	D	E	F	G	H
1		Agent	Sales	Expenses				
2		Jones	\$3,466	(\$82)				
3		Deacon	\$9,085	(\$100)				
4		Xi	\$7,997	(\$73)				
5		Omalpu	\$1,398	(\$53)				
6								
7		Agent	Total					
8		Deacon	\$8,985					

The formula is:

=SUMPRODUCT(((Table1[Sales])+Table1[Expenses]))*(Table1[Agent]=B8)), and it returns the sum of all sales and expenses for the agent listed in cell B8.

Example 3

In this example, we want to return the total of a particular item sold by a given region. In this case, how many cherries did the East region sell?

D12		=SUMPRODUCT((B2:B9=B12)*(C2:C9=C12)*D2:D9)				
	A	B	C	D	E	F
1		Region	Item	Sales		
2		North	Apples	\$2,763		
3		South	Pears	\$9,359		
4		East	Cherries	\$3,830		
5		West	Bananas	\$8,720		
6		North	Pears	\$1,873		
7		South	Apples	\$4,065		
8		East	Cherries	\$1,419		
9		West	Bananas	\$7,173		
10						
11		Region	Item	Sales		
12		East	Cherries	\$5,249		
13						

Here, the formula is: `=SUMPRODUCT((B2:B9=B12)*(C2:C9=C12)*D2:D9)`. It first multiplies the number of occurrences of East by the number of matching occurrences of cherries. Finally, it sums the values of the corresponding rows in the Sales column. To see how Excel calculates this, select the formula cell, then go to Formulas > Evaluate Formula > Evaluate.

3. Statistics-related Function (MIN, MAX, MEDIAN, etc)

● Min Function

Returns the smallest number in a set of values.

Syntax

`MIN(number1, [number2], ...)`

The MIN function syntax has the following arguments:

Number1, number2, ... Number1 is optional, subsequent numbers are optional. 1 to 255 numbers for which you want to find the minimum value.

Remarks

1. Arguments can either be numbers or names, arrays, or references that contain numbers.
2. Logical values and text representations of numbers that you type directly into the list of arguments are counted.
3. If an argument is an array or reference, only numbers in that array or reference are used. Empty cells, logical values, or text in the array or reference are ignored.
4. If the arguments contain no numbers, MIN returns 0.
5. Arguments that are error values or text that cannot be translated into numbers cause errors.
6. If you want to include logical values and text representations of numbers in a reference as part of the calculation, use the MINA function.

Copy the example data in the following table, and paste it in cell A1 of a new Excel worksheet. For formulas to show results, select them, press F2, and then press Enter. If you need to, you can adjust the column widths to see all the data.

Data
10
7
9
27
2

Formula	Description	Result
=MIN(A2:A6)	Smallest of the numbers in the range A2:A6.	2
=MIN(A2:A6,0)	Smallest of the numbers in the range A2:A6 and 0.	0

● Max Function

Returns the largest value in a set of values.

Syntax

MAX(number1, [number2], ...)

The MAX function syntax has the following arguments:

Number1, number2, ... Number1 is required, subsequent numbers are optional. 1 to 255 numbers for which you want to find the maximum value.

Remarks

1. Arguments can either be numbers or names, arrays, or references that contain numbers.
2. Logical values and text representations of numbers that you type directly into the list of arguments are counted.
3. If an argument is an array or reference, only numbers in that array or reference are used. Empty cells, logical values, or text in the array or reference are ignored.
4. If the arguments contain no numbers, MAX returns 0 (zero).
5. Arguments that are error values or text that cannot be translated into numbers cause errors.

6. If you want to include logical values and text representations of numbers in a reference as part of the calculation, use the MAXA function.

Example

Copy the example data in the following table, and paste it in cell A1 of a new Excel worksheet. For formulas to show results, select them, press F2, and then press Enter. If you need to, you can adjust the column widths to see all the data.

Data		
10		
7		
9		
27		
2		
Formula	Description	Result
=MAX(A2:A6)	Largest value in the range A2:A6.	27
=MAX(A2:A6, 30)	Largest value in the range A2:A6 and the value 30.	30

● Median Function

Returns the median of the given numbers. The median is the number in the middle of a set of numbers.

Syntax

MEDIAN(number1, [number2], ...)

The MEDIAN function syntax has the following arguments:

Number1, number2, ... Number1 is required, subsequent numbers are optional. 1 to 255 numbers for which you want the median.

Remarks

1. If there is an even number of numbers in the set, then MEDIAN calculates the average of the two numbers in the middle. See the second formula in the example.

2. Arguments can either be numbers or names, arrays, or references that contain numbers.
3. Logical values and text representations of numbers that you type directly into the list of arguments are counted.
4. If an array or reference argument contains text, logical values, or empty cells, those values are ignored; however, cells with the value zero are included.
5. Arguments that are error values or text that cannot be translated into numbers cause errors.

Example

Copy the example data in the following table, and paste it in cell A1 of a new Excel worksheet. For formulas to show results, select them, press F2, and then press Enter. If you need to, you can adjust the column widths to see all the data.

Data		
1		
2		
3		
4		
5		
6		
Formula	Description	Result
=MEDIAN(A2:A6)	Median of the 5 numbers in the range A2:A6. Because there are 5 values, the third is the median.	3
=MEDIAN(A2:A7)	Median of the 6 numbers in the range A2:A7. Because there are six numbers, the median is the midway point between the third and fourth numbers.	3.5

● Average Function

Returns the average (arithmetic mean) of the arguments. For example, if the range A1:A20 contains numbers, the formula =AVERAGE(A1:A20) returns the average of those numbers.

Syntax

AVERAGE(number1, [number2], ...)

The AVERAGE function syntax has the following arguments:

Number1 Required. The first number, cell reference, or range for which you want the average.

Number2, ... Optional. Additional numbers, cell references or ranges for which you want the average, up to a maximum of 255.

Remarks

1. Arguments can either be numbers or names, ranges, or cell references that contain numbers.
2. Logical values and text representations of numbers that you type directly into the list of arguments are not counted.
3. If a range or cell reference argument contains text, logical values, or empty cells, those values are ignored; however, cells with the value zero are included.
4. Arguments that are error values or text that cannot be translated into numbers cause errors.
5. If you want to include logical values and text representations of numbers in a reference as part of the calculation, use the AVERAGEA function.
6. If you want to calculate the average of only the values that meet certain criteria, use the AVERAGEIF function or the AVERAGEIFS function.

Example

Copy the example data in the following table, and paste it in cell A1 of a new Excel worksheet. For formulas to show results, select them, press F2, and then press Enter. If you need to, you can adjust the column widths to see all the data.

Data		
10	15	32
7		
9		
27		
2		

Formula	Description	Result
=AVERAGE(A2:A6)	Average of the numbers in cells A2 through A6.	11
=AVERAGE(A2:A6, 5)	Average of the numbers in cells A2 through A6 and the number 5.	10
=AVERAGE(A2:C2)	Average of the numbers in cells A2 through C2.	19

● Mode Function

returns the most frequently occurring, or repetitive, value in an array or range of data.

Syntax

MODE(number1,[number2],...)

The MODE function syntax has the following arguments:

Number1 Required. The first number argument for which you want to calculate the mode.

Number2,... Optional. Number arguments 2 to 255 for which you want to calculate the mode. You can also use a single array or a reference to an array instead of arguments separated by commas.

Remarks

- Arguments can either be numbers or names, arrays, or references that contain numbers.

2. If an array or reference argument contains text, logical values, or empty cells, those values are ignored; however, cells with the value zero are included.
3. Arguments that are error values or text that cannot be translated into numbers cause errors.
4. If the data set contains no duplicate data points, MODE returns the #N/A error value.

Example

Copy the example data in the following table, and paste it in cell A1 of a new Excel worksheet. For formulas to show results, select them, press F2, and then press Enter. If you need to, you can adjust the column widths to see all the data.

Data		
5.6		
4		
4		
3		
2		
4		
Formula	Description	Result
=MODE(A2:A7)	Mode, or most frequently occurring number above	4

4. VLOOKUP & HLOOKUP

● VLOOKUP function

Use VLOOKUP when you need to find things in a table or a range by row. For example, look up a price of an automotive part by the part number, or find an employee name based on their employee ID.

In its simplest form, the VLOOKUP function says:

=VLOOKUP(What you want to look up, where you want to look for it, the column number in the range containing the value to return, return an Approximate or Exact match – indicated as 1/TRUE, or 0/FALSE).

How to get started

There are four pieces of information that you will need in order to build the VLOOKUP syntax:

1. The value you want to look up, also called the lookup value.
2. The range where the lookup value is located. Remember that the lookup value should always be in the first column in the range for VLOOKUP to work correctly. For example, if your lookup value is in cell C2 then your range should start with C.
3. The column number in the range that contains the return value. For example, if you specify B2:D11 as the range, you should count B as the first column, C as the second, and so on.
4. Optionally, you can specify TRUE if you want an approximate match or FALSE if you want an exact match of the return value. If you don't specify anything, the default value will always be TRUE or approximate match.

Now put all of the above together as follows:

=VLOOKUP(lookup value, range containing the lookup value, the column number in the range containing the return value, Approximate match (TRUE) or Exact match (FALSE)).

Examples

Here are a few examples of VLOOKUP:

Example 1

	A	B	C	D	E
1	ID	Last name	First name	Title	Birth date
2	101	Davis	Sara	Sales Rep	12/08/68
3	102	Fontana	Olivier	VP (Sales)	02/19/52
4	103	Leal	Karina	Sales Rep	08/30/63
5	104	Patten	Michael	Sales Rep	09/19/58
6	105	Burke	Brian	Sales Manager	03/04/55
7	106	Sousa	Luis	Sales Rep	07/02/63
8					
9					
10	Formula	=VLOOKUP(B3,B2:E7,2,FALSE)			
11	Result	Olivier			
12					

VLOOKUP looks for *Fontana* in the first column (column B) in table_array B2:E7, and returns *Olivier* from the second column (column C) of the table_array. FALSE returns an exact match.

Example 2

	A	B	C	D	E
1	ID	Last name	First name	Title	Birth date
2	101	Davis	Sara	Sales Rep	12/08/68
3	102	Fontana	Olivier	VP (Sales)	02/19/52
4	103	Leal	Karina	Sales Rep	08/30/63
5	104	Patten	Michael	Sales Rep	09/19/58
6	105	Burke	Brian	Sales Manager	03/04/55
7	106	Sousa	Luis	Sales Rep	07/02/63
8					
9					
10	Formula	=VLOOKUP(102,A2:C7,2,FALSE)			
11	Result	Fontana			

VLOOKUP looks for an exact match (FALSE) of the last name for 102 (lookup_value) in the second column (column B) in the A2:C7 range, and returns *Fontana*.

Example 3

	A	B	C	D	E
1	ID	Last name	First name	Title	Birth date
2	101	Davis	Sara	Sales Rep	12/08/68
3	102	Fontana	Olivier	VP (Sales)	02/19/52
4	103	Leal	Karina	Sales Rep	08/30/63
5	104	Patten	Michael	Sales Rep	09/19/58
6	105	Burke	Brian	Sales Manager	03/04/55
7	106	Sousa	Luis	Sales Rep	07/02/63
8					
9					
10	Formula	=IF(VLOOKUP(103,A1:E7,2,FALSE)="Sousa","Located","Not found")			
11	Result	Not found			

IF checks to see if VLOOKUP returns *Sousa* as the last name of employee corresponding to 103 (lookup_value) in A1:E7 (table_array). Because the last name corresponding to 103 is *Leal*, the IF condition is false, and *Not found* is displayed.

Example 4

	A	B	C	D	E
1	ID	Last name	First name	Title	Birth date
2	101	Davis	Sara	Sales Rep	12/08/68
3	102	Fontana	Olivier	VP (Sales)	02/19/52
4	103	Leal	Karina	Sales Rep	08/30/63
5	104	Patten	Michael	Sales Rep	09/19/58
6	105	Burke	Brian	Sales Manager	03/04/55
7	106	Sousa	Luis	Sales Rep	07/02/63
8					
9					
10	Formula	=INT(YEARFRAC(DATE(2014,6,30), VLOOKUP(105,A2:E7,5, FALSE), 1))			
11	Result	59			
12					
13					
14					
15					

VLOOKUP looks for the birth date of the employee corresponding to 105 (lookup_value) in the A2:E7 range (table_array), and returns 03/04/1955. Then, YEARFRAC subtracts this birth date from 2014/6/30 and returns a value, which is then converted by INT to the integer 59.

Example 5

	A	B	C	D	E
1	ID	Last name	First name	Title	Birth date
2	101	Davis	Sara	Sales Rep	12/08/68
3	102	Fontana	Olivier	VP (Sales)	02/19/52
4	103	Leal	Karina	Sales Rep	08/30/63
5	104	Patten	Michael	Sales Rep	09/19/58
6	105	Burke	Brian	Sales Manager	03/04/55
7	106	Sousa	Luis	Sales Rep	07/02/63
8					
9					
10	Formula	=IF(ISNA(VLOOKUP(105,A2:E7,2,FALSE)) = TRUE, "Employee not found", VLOOKUP(105,A2:E7,2,FALSE))			
11	Result	Burke			
12					
13					
14					

IF checks to see if VLOOKUP returns a value for last name from column B for 105 (lookup_value). If VLOOKUP finds a last name, then IF will display the last name, otherwise IF returns *Employee not found*. ISNA makes sure that if VLOOKUP returns #N/A, then the error is replaced by *Employee not found*, instead of #N/A.

In this example, the return value is *Burke*, which is the last name corresponding to 105.

● HLOOKUP function

Searches for a value in the top row of a table or an array of values, and then returns a value in the same column from a row you specify in the table or array. Use HLOOKUP when your comparison values are located in a row across the top of a table of data, and you want to look down a specified number of rows. Use VLOOKUP when your comparison values are located in a column to the left of the data you want to find.

The H in HLOOKUP stands for "Horizontal."

Syntax

HLOOKUP(lookup_value, table_array, row_index_num, [range_lookup])

The HLOOKUP function syntax has the following arguments:

1. **Lookup_value** Required. The value to be found in the first row of the table. Lookup_value can be a value, a reference, or a text string.
2. **Table_array** Required. A table of information in which data is looked up. Use a reference to a range or a range name.
 - The values in the first row of table_array can be text, numbers, or logical values.
 - If range_lookup is TRUE, the values in the first row of table_array must be placed in ascending order: ...-2, -1, 0, 1, 2,... , A-Z, FALSE, TRUE; otherwise, HLOOKUP may not give the correct value. If range_lookup is FALSE, table_array does not need to be sorted.
 - Uppercase and lowercase text are equivalent.
 - Sort the values in ascending order, left to right. For more information, see Sort data in a range or table.
3. **Row_index_num** Required. The row number in table_array from which the matching value will be returned. A row_index_num of 1 returns the first row value in table_array, a row_index_num of 2 returns the second row value in table_array, and so on. If row_index_num is less than 1, HLOOKUP returns the #VALUE! error value; if row_index_num is greater than the number of rows on table_array, HLOOKUP returns the #REF! error value.
4. **Range_lookup** Optional. A logical value that specifies whether you want HLOOKUP to find an exact match or an approximate match. If TRUE or omitted, an approximate match is returned. In other words, if an exact match is not found, the next largest value that is less than lookup_value is returned. If FALSE, HLOOKUP will find an exact match. If one is not found, the error value

#N/A is returned.

Remark

1. If HLOOKUP can't find lookup_value, and range_lookup is TRUE, it uses the largest value that is less than lookup_value.
2. If lookup_value is smaller than the smallest value in the first row of table_array, HLOOKUP returns the #N/A error value.
3. If range_lookup is FALSE and lookup_value is text, you can use the wildcard characters, question mark (?) and asterisk (*), in lookup_value. A question mark matches any single character; an asterisk matches any sequence of characters. If you want to find an actual question mark or asterisk, type a tilde (~) before the character.

Example

Copy the example data in the following table, and paste it in cell A1 of a new Excel worksheet. For formulas to show results, select them, press F2, and then press Enter. If you need to, you can adjust the column widths to see all the data.

Axles	Bearings	Bolts
4	4	9
5	7	10
6	8	11

Formula	Description	Result
=HLOOKUP("Axles", A1:C4, 2, TRUE)	Looks up "Axles" in row 1, and returns the value from row 2 that's in the same column (column A).	4
=HLOOKUP("Bearings", A1:C4, 3, FALSE)	Looks up "Bearings" in row 1, and returns the value from row 3 that's in the same column (column B).	7
=HLOOKUP("B", A1:C4, 3, TRUE)	Looks up "B" in row 1, and returns the value from row 3 that's in the same column. Because an exact match for "B" is not found, the largest value in row 1 that is less than "B" is used: "Axles," in column A.	5
=HLOOKUP("Bolts", A1:C4, 4)	Looks up "Bolts" in row 1, and returns the value from row 4 that's in the same column (column C).	11
=HLOOKUP(3, {1,2,3;"a","b","c";"d","e","f"}, 2, TRUE)	Looks up the number 3 in the three-row array constant, and returns the value from row 2 in the same (in this case, third) column. There are three rows of values in the array constant, each row separated by a semicolon (;). Because "c" is found in row 2 and in the same column as 3, "c" is returned.	c

Reference :

1. <https://support.microsoft.com/en-us/office/if-function-69aed7c9-4e8a-4755-a9bc-aa8bbff73be2#:~:text=The%20IF%20function%20is%20one,if%20your%20comparison%20is%20False.>
2. <https://support.microsoft.com/en-us/office/ifs-function-36329a26-37b2-467c-972b-4a39bd951d45>
3. <https://support.microsoft.com/en-us/office/sum-function-043e1c7d-7726-4e80-8f32-07b23e057f89>
4. <https://support.microsoft.com/en-us/office/sumif-function-169b8c99-c05c-4483-a712-1697a653039b>
5. <https://support.microsoft.com/en-us/office/sumproduct-function-16753e75-9f68-4874-94ac-4d2145a2fd2e>
6. <https://support.microsoft.com/en-us/office/min-function-61635d12-920f-4ce2-a70f-96f202dcc152>
7. <https://support.microsoft.com/en-us/office/min-function-61635d12-920f-4ce2-a70f-96f202dcc152>
8. <https://support.microsoft.com/en-us/office/max-function-e0012414-9ac8-4b34-9a47-73e662c08098>

9. <https://support.microsoft.com/en-us/office/median-function-d0916313-4753-414c-8537-ce85bdd967d2>
10. <https://support.microsoft.com/en-us/office/average-function-047bac88-d466-426c-a32b-8f33eb960cf6>
11. <https://support.microsoft.com/en-us/office/mode-function-e45192ce-9122-4980-82ed-4bdc34973120>
12. <https://support.microsoft.com/en-us/office/vlookup-function-0bbc8083-26fe-4963-8ab8-93a18ad188a1>
13. <https://support.microsoft.com/en-us/office/hlookup-function-a3034eec-b719-4ba3-bb65-e1ad662ed95f>